mała delta

Ujarzmianie złożoności

Andrzej WALAT

Andrzej Walat, Ośrodek Edukacji Informatycznej i Zastosowań Komputerów w Warszawie

Logo jest środowiskiem, w którym można znakomicie uczyć się, jak radzić sobie ze złożonością. W tym artykule omówię trzy rozwiązania zadania o wadze szalkowej z finału Mazowieckiego Konkursu Informatycznego 2000 (dla uczniów dawnej szkoły podstawowej) i na tym przykładzie przedstawię techniki ujarzmiania złożoności.

Zadanie o wadze szalkowej

Jeśli mamy po jednym odważniku o masie 1 kg, 3 kg, 9 kg itd. aż do 3^n kg, to możemy na wadze szalkowej odważyć każdy ciężar, którego masa w kg jest liczbą całkowitą z zakresu od 0 do $1 + 3 + \ldots + 3^n$ i to w dodatku tylko w jeden sposób. Na przykład, żeby zrównoważyć ciężar 11 kg, trzeba na jednej szali położyć dany ciężar i odważnik 1kg, a na drugiej szali – odważniki 9 kg i 3 kg.

Zdefiniuj funkcję `ZRW :mc`, która dla danej masy ciężaru, będącej liczbą całkowitą dodatnią, wyznacza odważniki, które trzeba położyć na szalach wagi, żeby zrównoważyć dany ciężar. Wynikiem funkcji ma być dwuelementowa lista. Pierwszym jej elementem powinna być lista mas odważników, które należy położyć na jednej szali z danym ciężarem, a drugim elementem – lista mas odważników, które należy położyć na przeciwnej szali. Obie te listy powinny być uporządkowane malejąco.

Przykłady

Poprawnym wynikiem wyrażenia:	jest:
`ZRW 101`	`[[9 1] [81 27 3]]`
`ZRW 282`	`[[] [243 27 9 3]]`

Rozwiązanie 1

Używając 3 odważników o masie 1, 3 oraz 9 kg, można zrównoważyć na wadze szalkowej każdy ciężar, którego masa `mc` w kg jest liczbą całkowitą z zakresu od 0 do 13 włącznie. Na przykład:

- ciężar 10 kg można zrównoważyć, kładąc na przeciwnej szali odważniki [9 1],
- ciężar 11 kg można zrównoważyć, kładąc na tej samej szali odważnik 1, a na przeciwnej [9 3].

Największy ciężar, jaki można zrównoważyć za pomocą czterech odważników 1, 3, 9, 27, to $1 + 3 + 9 + 27 = 40$. Ale czy każdy ciężar o masie `mc` od 0 do 40 da się odważyć, jeśli mamy tylko te 4 odważniki?

- Każdy od 0 do 13 kg tak, bo wtedy wystarczą 3 odważniki: 1, 3 oraz 9.
- Każdy od 14 do 27 kg tak, bo po położeniu na przeciwnej szali odważnika 27 kg trzeba będzie zrównoważyć $27 - \texttt{mc} \leqslant 13$; a to można osiągnąć, używając odważników 1, 3 oraz 9.
- Każdy od 28 do 40 kg tak, po położeniu na przeciwnej szali odważnika 27 kg i wtedy trzeba będzie zrównoważyć $\texttt{mc} - 27 \leqslant 13$ kg; a to można osiągnąć, używając odważników 1, 3 oraz 9.

Podobnie można wykazać, że za pomocą 5 odważników 1, 3, 9, 27, 81 można zrównoważyć na wadze szalkowej każdy ciężar, którego masa `mc` w kg jest liczbą całkowitą nie większą niż 121, itd.

Problem zrównoważenia ciężaru o danej masie jest prosty, jeśli ta masa jest nieduża, np. nie większa niż 13 albo jeszcze lepiej 4. Jednak jeśli masa ciężaru jest bardzo duża, problem staje się złożony. Sposobem na złożoność jest **redukcja złożoności**. Problem zrównoważenia ciężaru o dużej masie można zredukować do problemu zrównoważenia ciężaru o odpowiednio mniejszej masie. Trzeba w tym celu znaleźć najcięższy (maksymalny) odważnik `mo`, jakiego musimy użyć, by zrównoważyć dany ciężar (to nie jest takie trudne) i położyć go na przeciwnej szali.

- Jeśli $\texttt{mc} = \texttt{mo}$, to koniec, ciężar jest zrównoważony.
- Jeśli $\texttt{mc} > \texttt{mo}$, to trzeba jeszcze zrównoważyć mniejszą masę $\texttt{mc} - \texttt{mo}$.

- Jeśli `mo` > `mc`, to trzeba zrównoważyć mniejszą masę `mo` – `mc`, nadmiar na przeciwnej szali.

Ten pomysł redukcji złożonego przypadku do prostszego został wykorzystany w rozwiązaniu, na które składają się 4 procedury przedstawione na poniższym wydruku.

```
OTO ZRW :mc
  WYNIK BALANS :mc MODW :mc
JUŻ

OTO BALANS :mc :mo
  JEŚLI :mc = :mo [WYNIK LISTA [] ( LISTA :mo )]
  JEŚLI :mc > :mo [WYNIK DOPW :mo ZRW :mc - :mo]
  WYNIK DOPW :mo WSPAK ZRW :mo - :mc
JUŻ
```

```
OTO MODW :mc
  NIECH "PTG3 1
  NIECH "SPTG3 1
  DOPÓKI [:mc > :SPTG3][PRZYP "PTG3 3 * :PTG3
  PRZYP "SPTG3 :SPTG3 + :PTG3]
  WYNIK :PTG3
JUŻ

OTO DOPW :EL :PARA
  WYNIK LISTA PIERW :PARA NAP :EL OST :PARA
JUŻ
```

Pozioma kreska oddziela dwie podstawowe procedury określające ogólny plan działań oraz dwie pomocnicze procedury – środki realizacji planu.

Procedura `BALANS` określa sposób redukcji zagadnienia zrównoważenia ciężaru do prostszego zagadnienia zrównoważenia mniejszego ciężaru. Musi ona mieć dwie dane: `mc` – masę ciężaru oraz `mo` – masę maksymalnego odważnika, jakiego trzeba użyć, by zrównoważyć dany ciężar.

Funkcja – `ZRW` znajduje wynik, wywołując `BALANS` z dwiema danymi: `mc` – masa ciężaru oraz `MODW :mc` – masa maksymalnego odważnika, jakiego trzeba użyć, by zrównoważyć dany ciężar.

Trzeba jeszcze wyjaśnić, w jaki sposób funkcja `MODW` wyznacza swoją wartość – masę maksymalnego odważnika, jakiego trzeba użyć, by zrównoważyć dany ciężar. Wyjaśnię to na przykładzie. Załóżmy, że chcemy wyznaczyć masę maksymalnego odważnika, jakiego trzeba użyć, by zrównoważyć ciężar o masie 101 kg. W tym celu będziemy obliczali kolejne potęgi liczby 3 (wartości `PTG3`) oraz kolejne sumy potęg liczby 3 (wartości `SPTG3`) tak długo, dopóki dana masa 101 będzie większa niż wartość `SPTG3`.

PTG3	SPTG3
1	1
3	4
9	13
27	40
81	121

Końcowa wartość `PTG3` – w tym przypadku 81 – to szukany maksymalny odważnik. W taki sam sposób funkcja `MODW` znajduje wynik w dowolnym przypadku. Trzeba dodać, że funkcję `MODW`, jak prawie wszystko, można zdefiniować na wiele sposobów. Można np. wykorzystać spostrzeżenie, że kolejna wartość `SPTG3` – to zawsze potrojona poprzednia wartość plus 1.

```
OTO MODW :mc
  NIECH "SPTG3 1
  DOPÓKI [:mc > :SPTG3][ PRZYP "SPTG3 3 * :SPTG3
  + 1]
  WYNIK :SPTG3 - (:SPTG3 - 1 ) / 3
JUŻ
```

Funkcję `MODW` możemy zdefiniować jak tylko nam się spodoba (byle poprawnie), nie ma to żadnego wpływu na nasz ogólny plan – treść funkcji `ZRW` oraz `BALANS`. Tak samo, jak zmiana elementu złożonej konstrukcji, np. okien, nie wpływa na ogólny plan budowli (pod warunkiem, że elementy zachowują ustalone parametry, np. rozmiary). Taka separowalność części rozwiązania zadania jest cechą dobrego programowania.

Ostatnia z czterech funkcji składających się na rozwiązanie `DOPW` musi mieć dwie dane: jakiś nowy element `EL` oraz listę dwóch list `PARA`. Jej wynikiem jest dana lista dwóch list z nowym elementem dopisanym na początek drugiej listy.

Rozwiązanie 2

Prawie każde zadanie można rozwiązać na 1001 różnych sposobów. Myśl o tym jest pocieszająca. Jeśli może być tak wiele różnych pomysłów rozwiązania zadania, dlaczego przynajmniej jeden z nich nie miałby wpaść do mojej głowy?

Skorzystamy z rady pochodzącej z książeczki George'a Polyi *Jak to rozwiązać*.

„Jeśli nie możesz rozwiązać postawionego zadania, spróbuj najpierw rozwiązać jakieś zadanie pokrewne. Czy nie mógłbyś wymyślić bardziej dostępnego zadania pokrewnego?"

Owszem, moglibyśmy. Moglibyśmy, na przykład, założyć, że dysponujemy nie po jednym tylko odważniku o masie 1, 3, 9, 27,... kg, ale mamy po dwa odważniki każdego rodzaju. Nietrudno zauważyć, że w takim przypadku każdy ciężar, którego masa `mc` w kg jest nieujemną liczbą całkowitą, można zrównoważyć, kładąc odważniki wyłącznie na przeciwną szalę niż ciężar. Żeby je wyznaczyć, wystarczy znaleźć zapis liczby `mc` w systemie trójkowym.

Na przykład liczba $667 = 2 \cdot 243 + 2 \cdot 81 + 0 \cdot 27 + +2 \cdot 9 + 0 \cdot 3 + 1 \cdot 1$ ma zapis trójkowy 220201. Oznacza to, że ciężar o masie 667 kg można zrównoważyć, kładąc na przeciwnej szali: 2 odważniki 243 kg, 2 odważniki 81 kg, dwa odważniki 9 kg oraz 1 odważnik 1 kg. Ten przykład pokazuje, jak łatwo można rozwiązać wymyślone przez nas pokrewne zadanie. No dobrze, ale czy można to wykorzystać do rozwiązania naszego oryginalnego zadania? Tak, i to bardzo prosto. Popatrzmy na zapis zrównoważenia ciężaru 667 kg w postaci następującego rysunku-tabelki:

2	2	0	2	0	1
243	81	27	9	3	1
0	0	0	0	0	0

Górny wiersz tabeli wskazuje, jakie odważniki leżą na przeciwnej szali niż ciężar, a dolny – jakie razem z ciężarem (aktualnie żadne). Jest to dobre rozwiązanie pokrewnego zadania, ale nie zadania oryginalnego, ponieważ nie dysponujemy dwoma odważnikami o masie 9 kg ani 81 kg. Możemy za to kłaść odważniki także na szalce razem z ciężarem. Zamiast 2 odważników 9 kg, możemy położyć na przeciwnej szalce niż ciężar 1 odważnik 27 kg oraz 1 odważnik 9 kg na szalce razem z ciężarem. W ten sposób otrzymamy następującą sytuację równowagi:

2	2	1	0	0	1
243	81	27	9	3	1
0	0	0	1	0	0

Teraz trzeba coś zrobić z dwoma odważnikami 81 kg na szalce przeciwnej niż ciężar. Podobnie jak poprzednio, zdejmiemy je, a za to dołożymy na tej samej szali 1 odważnik o masie 243 kg oraz razem z ciężarem 1 odważnik o masie 81 kg:

3	0	1	0	0	1
243	81	27	9	3	1
0	1	0	1	0	0

W ten sposób milcząco zakładamy, że pożyczyliśmy skądś trzeci odważnik 243 kg, ale tylko na chwilę, zaraz go oddamy. Zdejmiemy z szalki trzy odważniki 243 kg i zamiast tego położymy 1 odważnik 729 kg:

1	0	0	1	0	0	1
729	243	81	27	9	3	1
0	0	1	0	1	0	0

Teraz mamy sytuację równowagi i każdy odważnik od 1 do 729 został użyty co najwyżej 1 raz. Na jednej szalce mamy ciężar 667 kg i odważniki o łącznej masie 81 kg + 9 kg = 90 kg, razem 757 kg. Na drugiej szali: 729 kg + 27 kg + 1 kg.

Ten sposób postępowania, zilustrowany przykładem, nietrudno zapisać w postaci odpowiednich procedur w Logo znajdujących szukane zrównoważenie dowolnego ciężaru. Można go także na wiele sposobów udoskonalić. Tego jednak tutaj nie będziemy robili, pozostawiając to zadanie zainteresowanym Czytelnikom. Czasem warto przeanalizować różne możliwe idee rozwiązania problemu nie po to, żeby je do końca wykorzystać, zrealizować w postaci procedur w jakimś języku programowania, ale po to, żeby lepiej, wszechstronniej zrozumieć problem i być może dzięki temu odkryć inne pomysły.

Rozwiązanie 3

W tym przypadku także skorzystamy z rady George'a Polyi.

„Jeśli nie możesz rozwiązać postawionego zadania, to … czy nie mógłbyś wymyślić jakiegoś bardziej ogólnego zadania?”

Zdefiniujemy funkcję `BALANS`, która musi mieć dwie dane:

• dowolną potęgę `p` liczby 3 oraz

• masę ciężaru `mc` daną jako wielokrotność `p`,

a jej wynikiem jest zrównoważenie ciężaru określone w postaci listy dwóch list, tak jak tego wymagamy od funkcji `ZRW`.

Funkcja `BALANS` jest uogólnieniem `ZRW`, bo wynik `ZRW` dla dowolnej całkowitej masy ciała `mc` można otrzymać jako wynik `BALANS` z ustaloną wartością drugiego parametru 1:

```
OTO ZRW :mc
  WYNIK BALANS :mc 1
JUŻ
```

Załóżmy, że mamy daną masę ciężaru będącą wielokrotnością `mc` potęgi `p` liczby 3.

Jak w takim przypadku znaleźć odpowiednie zrównoważenie (balans)? Oczywiście w szczególnie prostym przypadku, gdy `mc` = 0, balans osiągamy, nie kładąc na obu szalach nic; wynikiem powinna być lista dwóch pustych list: `[[][]]`.

W bardziej złożonym przypadku, gdy `mc` nie jest zerem, możemy obliczyć resztę z dzielenia `mc` przez 3 i zredukować odpowiednio problem, zależnie od otrzymanej reszty.

Jeśli reszta = 0, to zadanie redukuje się do znalezienia wyniku:
`BALANS ILORAZC :MC 3 3 * :P`.

Jeśli reszta = 1, to trzeba znaleźć `BALANS ILORAZC :MC 3 3 * :P` – jest to lista dwóch list – i do prawej z tych dwóch list dopisać na koniec `:p`.

Gdy reszta = 2, to trzeba położyć odważnik o masie `p` razem z ciężarem i znaleźć zrównoważenie odpowiednio większego o p ciężaru. Trzeba znaleźć `BALANS ( ILORAZC :MC 3 ) + 1 3 * :P` – jest to lista dwóch list i do lewej z tych dwóch list dopisać na koniec `:p`.

Pełnym rozwiązaniem zadania jest zestaw czterech procedur:

```
OTO ZRW :mc
  WYNIK BALANS :mc 1
JUŻ

OTO BALANS :mc :P
  JEŚLI :mc = 0 [WYNIK LISTA [] []]
  WYNIK WYBIERZ RESZTA :mc 3 [0 [BALANS ILORAZC
                         :mc 3 3 * :P]
                         1 [DOPW :P BALANS ILORAZC
                         :mc 3 3 * :P]
                         2 [DOLW :P BALANS ( ILORAZC
                         :mc 3 ) + 1 3 * :P]]
JUŻ
```

```
OTO DOPW :el :para
  WYNIK LISTA PIERW :para NAK :el OST :para
JUŻ

OTO DOLW :el :para
  WYNIK LISTA NAK :el PIERW :para OST :para
JUŻ
```

Podobnie, jak w rozwiązaniu 1, pierwsze dwie wytyczają ogólny plan postępowania, a dwie procedury pod kreską to pomocnicze środki realizacji.

W tym rozwiązaniu funkcja `DOPW` musi mieć dwie dane: jakiś nowy element `EL` oraz listę dwóch list `PARA`. Jej wynikiem jest dana para list z nowym elementem dopisanym na koniec drugiej listy.

Funkcja `DOLW` dopisuje nowy element na koniec pierwszej listy w danej parze list.

O znaczeniu grubej kreski – abstrakcja danych

O funkcjach `DOPW` oraz `DOLW` warto chwilę pomyśleć w oderwaniu od aktualnego zadania. Funkcje dopisujące nowy element do lewego lub prawego elementu danej pary list mogą być użyteczne w różnych sytuacjach. Mogłyby istnieć już w Logo jako funkcje pierwotne obok innych funkcji, takich jak `NAP` oraz `NAK`. Jest wiele dialektów Logo, więc może istnieje taki, w którym `DOPW` oraz `DOLW` są funkcjami pierwotnymi i nie trzeba ich definiować. Wyobraźmy sobie, że istnieje. W takim Logo rozwiązanie naszego zadania byłoby jeszcze prostsze i składałoby się tylko z dwóch krótkich definicji funkcji `ZRW` oraz `BALANS`. Za tym „wyobraźmy sobie” kryje się ważna technika radzenia sobie ze złożonością. Kiedy mam do rozwiązania złożony problem i wiem, że będę potrzebował pewnych narzędzi, których nie ma w konkretnym Logo, jakim dysponuję, mogę sobie wyobrazić, że je mam. Tworzę wtedy swoje rozwiązanie w **wirtualnym Logo**, w którym mam wszystkie potrzebne narzędzia. To bardzo upraszcza rozwiązanie problemu. Następnie, żeby móc uruchomić moje rozwiązanie, poprawiam Logo, dodając nowe środki „procedury pod kreską”. Ich zdefiniowanie jest już osobnym i zwykle prostym zadaniem. Technika, która polega na tym, że najpierw używam środków, którymi faktycznie nie dysponuję, jak bym je miał, i nie martwię się, jak je zdefiniuję, a następnie osobno definiuję potrzebne środki – nie zaprzątając sobie głowy, do czego są mi aktualnie potrzebne – nazywa się **abstrakcją danych**. Ważne cechy abstrakcji danych to:

1. Separowalność używania i definiowania. Używam czegoś, nie martwiąc się, jak to zdefiniuję. Definiuję, nie zaprzątając sobie głowy, do czego aktualnie chcę tego używać.

2. Tworzenie produktów wielorakiego użytku (*reusage*). Analizując różne problemy, wyodrębniamy środki (funkcje, procedury, typy struktur danych), które mogą mieć zastosowanie w różnorodnych sytuacjach; zwykle nadajemy im jakąś sensowną nazwę i próbujemy zdefiniować jako abstrakcyjne obiekty do wielorakiego użytku – w oderwaniu od aktualnej potrzeby.

Złożoność ujarzmiona

Każda epoka ma swoje wyzwania. Wielkim wyzwaniem na miarę XXI wieku jest ujarzmianie złożoności. Opanowanie złożoności wymaga czasem cierpliwego wielokrotnego podchodzenia do problemu z różnych stron, jak do dzikiego mustanga. Ale jej okiełznanie może dać równie wielką satysfakcję. Problem, który mógł się wydawać złożony i wymykał się spod kontroli, daje się nagle zamknąć w kilku prostych definicjach. To jest właśnie złożoność ujarzmiona.